

2.2 GENERACIÓN DE UN TREN DE PULSOS – “OUTPUT COMPARE”

USO DEL COMPARADOR DE SALIDA TIPO “UNBUFFERED”

	ÍNDICE	
Preparado por: Rangel Alvarado Estudiante Graduando de Lic. en Ing. Electromecánica Universidad Tecnológica de Panamá Panamá, Panamá “e-mail”: issaiass@cwpanama.net “web site”: http://www.geocities.com/issaiass/	2.2.1	Introducción 194
	2.2.2	Características de una Onda Cuadrada en Comparación de Salida 195
	2.2.3	Registros de Estado y Control de los Canales del TIM 195
	2.2.4	Interrupción del Canal 197
	2.2.5	Diagrama de Flujo 197
	2.2.6	Código 199
	2.2.7	Simulación 203
	2.2.8	Conclusión 204
	2.2.9	Referencias 205
	2.2.10	Problemas Propuestos 205

2.2.1 Introducción

A pesar de que el uso de el registro de control TSC aparenta ser muy completo, el mismo es muy limitado en aplicaciones de control, como por ejemplo una generación de trenes de pulsos. Para esto, el microcontrolador tiene más opciones de registro de control de temporizadores que hacen más fácil implementar estas funciones.

Se hará referencia en este documento a los canales del temporizador, específicamente a la función “Output Compare” cuya función puede ser:

- **Generar un tren de pulsos.**
- Indicar el tiempo transcurrido a un circuito externo.
- Disparo de eventos externos de tipo “Unbuffered”.

El objetivo de esta nota es:

- Estudiar las propiedades de una onda cuadrada: relacionar las propiedades de una onda con los registros internos del temporizador.
- Configurar el temporizador para generar un tren de pulsos tipo “Unbuffered”¹: generar una señal continua que pudiese servir como reloj a un circuito externo.
- Manejar la interrupción del canal del temporizador (“Output Compare”): ver el salto a la rutina de interrupción el cual posibilita hacer operaciones alternativas sin ocupar el tiempo del CPU.
- Simular: Utilizar “breakpoints” para depurar o ver la funcionalidad del código.

Nota: Se asume un dominio de la NT0101 y se hace referencia cruzada a la misma.

¹ Se le dice “Unbuffered” porque se reemplaza un nuevo valor de ancho de pulso cada vez que se renueva el conteo.

2.2.2 Características de una Onda Cuadrada en Comparación de Salida





Figura 119 (izquierda). Propiedades de una Onda Cuadrada en “Output Compare”. (a) Registro del Canal. El registro programable TCHX, genera un pulso periódico con una polaridad programada, duración y frecuencia. (b) Sobreflujo del Canal del Temporizador. Cada vez que se sobrepase la cuenta (TCNT[H:L]) con el registro del canal (TCHX[H:L]) se puede conmutar el estado del puerto, lo que genera un tren de pulsos. (c) Desplazamiento. Para obtener un tren de pulsos en modo de comparación de salida es necesario que se le sume el desplazamiento al registro del canal.

Nota: X = 0 ó 1. Canales del Temporizador.

2.2.3 Registros de Estado y Control de los Canales del TIM

2.2.3.1 Registros de Estado y Control de los Canales – TSC0, TSC1

- 0 = No ha ocurrido un desborde de conteo.
- 1 = Ocurrió un desborde de Conteo $TCHX[H:L] > TCNT[H:L]$.
- 0 = Interrupción del Canal Deshabilitada.
- 1 = Interrupción del Canal Habilitada.



Nota: Si $ELSXB[A] \neq 0$, la combinación MSXA es válida.
 MS1B no disponible, solo lee un cero.

Tabla 51. Modo de Comparación de Salida “Unbuffered”

Comparación de Salida [UnBuffered]	MSXB	MSXA	ELSXB	ELSXA
Conmuta en la comparación de salida	0	1	0	1
Borra en la comparación de salida	0	1	1	0
Impone uno en la comparación de salida	0	1	1	1

X = 0 ó 1. Canal de Temporización.

- MSXB = Modo de Selección del Canal X, Bit B. [Mode Select Bit B].
- MSXA = Modo de Selección del Canal X, Bit A. [Mode Select Bit A].
- ELSXB = Nivel de Selección del Canal X, Bit B. [Edge Level Select Bit B].
- ELSXA = Nivel de Selección del Canal X, Bit A. [Edge Level Select Bit A].

Figura 120 (izquierda). Registro de Estado y Control del Canal del TIM. Con la ayuda de la tabla 51, configure el registro para generar trenes de pulsos haciendo que conmute al comparar.

2.2.3.2 Registros del Canal – TCH0[H:L], TCH1[H:L]




Figura 121. Registros del Canal. El valor depositado en el registro del canal es el valor a comparar al inicializarse en modo de comparación de salida (¿TCHX[H:L] > TCNT[H:L]?)

Nota: El modo “Output Compare” queda inhibido, si al escribirse el byte alto (TCHXH), no se ha escrito el byte bajo (TCHXL).

2.2.3.3 Configuración de Comparación de Salida en Modo “Unbuffered”

Refiérase en esta sección a la NT0101 – Sección 2.1.3.

Configure el Temporizador para una Comparación de Salida por el Canal 1 y generar un tren de pulsos de cinco (5) milisegundos, PS de seis (6) y que su pulso inicie en alto, habilite las interrupciones del canal para sumar el siguiente pulso a conmutar.

$$TCH[H : L] = \frac{t \cdot f_{xtal}}{2^{2+PS}} = \frac{0.005 \cdot 4.9152 \times 10^6}{2^{2+6}} = 96_{10} = 60_{16}$$

R = Reservado
 = No implementado




Figura 122. Configuración de una Comparación de Salida “Unbuffered”. El temporizador, canal 1, es utilizado para generar un tren de pulsos por el puerto PTD5.

2.2.4 Interrupciones del Canal

En esta sección ver tabla 52.

Si en algún momento, la bandera se levanta (CHXF = 1) y la máscara está habilitada (CHXIE = 1), el temporizador puede generar un evento de interrupción.

Tabla 52. Vectores de Interrupción del Canal del Temporizador

Bandera	Máscara	Dirección	Vector (Dirección)
CH1F	CH1IE	FFF4	Canal 1 del TIM (Alto)
		FFF5	Canal 1 del TIM (Bajo)
CH0F	CH0IE	FFF6	Canal 0 del TIM (Alto)
		FFF7	Canal 0 del TIM (Bajo)

2.2.5 Diagrama de Flujo

El siguiente programa hace conmutar un LED conectado al puerto PTD5 a intervalos de cinco (5) milisegundos, con un período repetitivo de 100 Hz.




Figura 123. NT0102 – Timer – “Output Compare”. (a) Programa Principal. Inicializa el temporizador en modo de comparación de salida en modo “unbuffered”. (b) Reinicio del sistema. Al presionar “reset”, el sistema reinicia automáticamente.




Figura 124. NT0102 – “Timer” – “Output Compare” – Subrutinas. (a) Inicialización de “Output Compare Unbuffered”. Inicializa el temporizador en modo de comparación de salida en modo “unbuffered” para generar un tren de pulsos en la salida PTD5. (b) Interrupción. Se utiliza la misma interrupción para ver el sobreflujo del canal y el temporizador, interrupción nula.

2.2.6 Código

```
=====
; ARCHIVO      : NT0102 - Timer - Output Compare - 24 03 04.asm
; PROPÓSITO   : Genera por medio de una base de tiempo de 5ms el conmutado de
;              un LED (PTD5).
;              - Observar la configuración del Canal del temporizador.
;              - Generación de un Output Compare Unbuffered (tren de pulsos)
;              P.D.: Agregar una resistencia limitadora al PTD4 de 3.3k con
;              un LED con su cátodo a tierra.
; NOTA       :
;              1 - Observar el resultado de la temporización en el LED PTD5
;              2 - Observar el resultado del registro TIM en las localidades
;              TSC1, TCH1[H:L].
; REFERENCIA:
;              Advanced Information of MCU68HC908JK1, JK3, JL3...
;              http://www.freescale.com/files/microcontrollers/
;              doc/data_sheet/MC68HC08JL3.pdf
;              Pág. 109 - Descripción Funcional
;              Pág. 115 - Interrupción
;              Págs. 121 @ 125. - Información de registros
;
; LENGUAJE    : IN-LINE ASSEMBLER
;-----
; HISTORIAL
; DD MM AA
; 10 03 03 Creado.
; 05 09 04 Modificado.
;-----
; Pasos para iniciar el TIM:
;
; 1 - Definir del registro TSC o registro de
; control del TIM:
;   ¿Necesito o no interrupciones? (TOIE)
;   ¿Cuál que valor de divisor ajustar? (PS[2:0])
; 2 - Definir el registro TMOD[H:L] como módulo máximo de
; conteo, TMOD[H:L] = t*fxtal/(2^(2+PS))
; 3 - Habilitar Interrupciones globales (cli)
; 4 - Habilitar el temporizador (TSTOP = 0)
; 5 - Configurar el vector interrupción ($FFF2)
; 6 - Escribir el código de la interrupción y su retorno (rti)
;
; Pasos para iniciar el Canal del TIM
; 1 - Definir del registro TSCX registro de
; control del canal del TIM:
;   ¿Necesito o no interrupciones? (CHXIE)
;   ¿Input Capture, Output Compare, PWM?
;   ¿Inicio en bajo o en alto? (MSXB,MSXA,ELSXB,ELSXA)
; 2 - Definir el registro TCHX[H:L] como módulo de comparación
; TCHX[H:L] = D * TMOD[H:L], D = valor en porcentaje
; 3 - Configurar el vector interrupción ($FFF4 - CH1;
```

NT0102

Rev. 1 del 06.08.05

```

;          $FFF6 - CH0)
;          4 - Escribir el código de la interrupción y su retorno (rti)
;=====
;$SET  ICS08          ; ICS08 = 1, Vamos a simular en la pastilla
;                          ; la velocidad de simulación es menor en la
;                          ; PC.
$SETNOT ICS08        ; ICS08 = 0, Vamos a programar la pastilla
;                          ; la aplicación debe correr en tiempo real

$IF ICS08
MSEG5      equ $0003    ; 5 milisegundos de base de tiempo - Simulado
$ELSEIF
MSEG5      equ $0060    ; 5 milisegundos de base de tiempo
$ENDIF

;=====
;          Definiciones del Usuario
;=====
COPD      equ 0T        ; Bit 0 del registro CONFIG1
;MSEG5    equ $0060     ; 1 milisegundo de base de tiempo
BIT5      equ 5T        ; TSC, Bit de Inicio de Conteo, Bit 5 ON
CH1F      equ 7T        ; TSC1, Bit de Salida Comparada
TRST      equ %00010000; TSC, Bit de Reinicio de conteo, Bit 4 ON
TSTOP     equ %00100000; TSC, Bit de pausa, Bit 5 ON
PS2       equ %00000100; Preescalar del Temporizador, Bit 2
PS1       equ %00000010; Preescalar del Temporizador, Bit 1
CH1IE     equ %01000000; Habilito interrupciones, canal 1
MS1A      equ %00010000; Output Compare Tipo Unbuffered
ELS1A     equ %00000100; Bit de Selección de flanco, activa...
;                          ; la señal en uno lógico al comparar

;=====
;          Mapa de Memoria del Microcontrolador
;=====
;=====
;          Registro de Configuraciones
;=====
CONFIG1   equ $001F     ; Vectores de configuración

;=====
;          Registro de Temporizador
;=====
TSC       equ $0020     ; Dirección, registro de estado y control del TIM
TCNTH     equ $0021     ; TCNT, Registro almacenador de cuenta del módulo,
;                          ; registro alto.
TCNTL     equ $0022     ; TCNT, Registro almacenador de cuenta del módulo,
TSC1      equ $0028     ; Dirección, registro de estado y control del TIM Canal
;                          ; 1
TCH1H     equ $0029     ; TCH1H, Registro almacenador de comparación de la
;                          ; salida, registro alto.
TCH1L     equ $002A     ; TCH1L, Registro almacenador de comparación de la
;                          ; salida, registro bajo.

```

```

;=====
;
;                               Memoria FLASH
;=====
FLASH_START equ $EC00    ; Puntero - Mem.FLASH

;=====
;
;                               Vectores de Usuario
;=====
TIMCH1H    equ $FFF4    ; Vector de Sobreflujo del TIM, Canal 1(Alto)
RESET_VEC  equ $FFFE    ; Puntero del RESET

;=====
; OBJETIVO    : Inicio de Codif. del Ensam-
;              blador en Memoria FLASH.
;=====
                org FLASH_START    ; Inicio Mem. FLASH

;=====
; OBJETIVO    : Configura el TIM para generar
;              una rata de parpadeo de 5
;              milisegundos.
;=====
START
    rsp                ; Inic.Stack = $00ff
    bset COPD,CONFIG1  ; Desactiva watchdog
    clra               ; Borra A
    jsr INIT_TIM       ; Inicializa TIM
    cli                ; Habilita Interrupciones
    bclr BIT5,TSC      ; Inicia el temporizado (TSTOP = 0)
ESPERA wait           ; Espera la interrupción.
    bra ESPERA         ; Salta al modo de bajo consumo

;=====
; INIT_TIM    : Inicializa el TIM
; OBJETIVO    : Inicialización del tempori-
;              zador.
;              Base de tiempo de 5ms
; ENTRADA     : Ninguna
; SALIDA      : Ninguna
; REGISTROS   :
; AFECTADOS   : TSC, TCH1[H:L], TSC1
;=====
INIT_TIM
    ldhx #MSEG5        ; Programa H:X para cargar al módulo con 5ms de
                       ; retardo
    sthx TCH1H         ; Almacena .5 ms de retardo en el registro del canal
    mov #{TRST|TSTOP|PS2|PS1},TSC
                       ; Reseteo el temporizador, habilito interrupciones
                       ; PS[2:0] = 6, el temporizador se encuentra detenido
                       ; por defecto.
    mov #{CH1IE|MS1A|ELS1A},TSC1
                       ; Habilito Interrupciones del canal y configuro
    
```

```

; un Output Compare Unbuffered, conmuto al sobre
; flujo del temporizador el puerto.
rts ; retorna

;=====
; TIMCH1L : Interrupción del Canal 1 del
; TIM y sobreflujos simulada
; OBJETIVO : Reconocer las interrupciones
; ENTRADA : Ninguna
; SALIDA : Ninguna
; REGISTROS
; AFECTADOS : TSC1, TCH1H, TCH1L, H:X
;=====
TIMCH1L
    bclr CH1F,TSC1 ; Reconoce salida comparada.
    ldhx TCH1H ; Carga H:X con TCH1[H:L]
    aix #MSEG5 ; Le agrega 5 milisegundos de retardo
    sthx TCH1H ; Deposita en TCH1[H:L]
    rti ; retorna

;=====
; OBJETIVO : Inicializa el Vector de Reset
; Arranque del programa en la
; memoria Flash y búsqueda de la
; interrupción del canal del
; temporizador.
;=====
;===== Vectores del TIM, sobreflujo y comparación =====
    org TIMCH1H ; Inicio de los vectores del TIM
    dw TIMCH1L ; Desborde del contador del Canal
;===== Vector de Reinicio de Sistema =====
    org RESET_VEC ; Puntero Vec - RESET
    dw START ; al darse reset salta a Start

```

Listado 12. NT0102 – “Timer” – “Output Compare”. El programa responde a inicializar el temporizador, canal 1 para generar un “Output Compare Unbuffered” simplemente para generar un tren de pulsos que pudiese servir como señal a un circuito externo.

2.2.7 Simulación

Para el uso de esta sección se asume tener dominio de la NT0006 (Sección 1.6.4), NT0009 (Sección 1.9.5) y NT0101 (Sección 2.1.3).

```

=====
TIMCH1L
! bclr CH1F,TSC1
+ ldhx TCH1H
+ aix #MSEG5
+ sthx TCH1H
+ rti
    
```

(a) Sitúe un “breakpoint” en la siguiente parte del código.

Output Compare Unbuffered. (TCH1)

Memory Window 1 - Data Sp					
0000	2E	E3	UU	BC	00
0008	UU	UU	00	UU	UU
0010	UU	UU	UU	UU	UU
0018	UU	UU	00	00	UU
0020	26	00	00	FF	FF
0028	54	00	60	UU	UU

Reg. de Configuración del Canal 1 (TSC1)

(b) Corra paso a paso hasta el llamado de la subrutina INIT_TIM. Ejecute paso a paso y observe que registros varían y cual es su función.

La cuenta iguala al ancho de pulso de la onda, salta a la interrupción si está habilitada.

0010	UU	UU	UU	UU	UU
0018	UU	UU	00	00	UU
0020	06	00	60	FF	FF
0028	D4	00	60	UU	UU

(c) Escriba Go en la entrada de línea de comandos y observe cuando se genera la interrupción..

0000	2E	E3	UU	BC	00
0008	UU	UU	00	UU	UU
0010	UU	UU	UU	UU	UU
0018	UU	UU	00	00	UU
0020	06	03	C0	FF	FF
0028	D4	03	C0	UU	UU

Luego de diez (10) pulsaciones de cinco (5) milisegundos cada una.

(d) Nuevamente escriba Go y observe cuando la cuenta iguala a TCH1.

Figura 125. NT0102 – Simulación del Programa Principal. (a) “Breakpoint”. Sitúe un breakpoint para ver el cambio entre el ancho de pulso y el periodo del canal. (b) Registros del Canal del Temporizador. En la rutina de inicialización del canal observe la variación de los registros del temporizador. (c) Desborde del Canal. Al desborde del canal, si la interrupción está habilitada, se puede generar un evento de interrupción. (d) Desborde de diez (10) canalizaciones de trenes de pulsos. Al desborde de sobreflujo del canal, se procede a conmutar la salida del pin en este caso el canal 1.

- (a) Inicie WinIDE.
- (b) Cargue el archivo NT0102 – Timer – Output Compare - 27 05 04.asm.
- (c) Compile.
- (d) Entre al simulador.
- (e) Añada un (1) “breakpoint” según la figura 125(a).
- (f) Corra su programa paso a paso y en la sección de llamado a la rutina de inicialización del temporizador (INIT_TIM) observe la ejecución de cada instrucción.
- (g) Corra su programa y espere a la llegada del “breakpoint”. Cuando la cuenta iguala al registro del canal, se da una interrupción si esta está activada, figura 125(c).
- (h) Nuevamente corra su programa y observe que cuando la cuenta iguala al módulo, sucede una nueva interrupción, figura 125(d).
- (j) Si desea “quemar” su pastilla, revisar la NT0009, Sección 1.9.5.

Nota: Recuerde mover su compilación condicional si desea que su programa corra en tiempo real:

```
; $SET ICS08  
$SETNOT ICS08
```

2.2.8 Conclusión

El temporizador fue utilizado para generar un tren de pulsos, configurándolo en el modo “Output Compare” o Comparación de Salida, el cual permite añadir esta función. Se le dice “Output Compare”, pues para añadir un nuevo ancho de pulso se escribe directamente el canal de temporización.

Para fin demostrativo se hizo un tren de pulsos de cinco (5) milisegundos con y se utilizó la interrupción del canal, ayudado de los “breakpoints” para visualizar el desborde de conteos.

2.2.9 Referencias

2.2.9.1 Información Avanzada sobre el Microcontrolador

(a) http://www.freescale.com/files/microcontrollers/doc/data_sheet/MC68HC08JL3.pdf

Pág. 109 – Descripción de la Función de Comparación de Salida, Sección 10.5.3

Pág. 115 – Interrupción de desborde del Canal, Sección 10.6, párrafo 2.

Págs. 117 – 121. - Información de registros de Control del Canal

2.2.9.2 Información Avanzada sobre el Microcontrolador

(a) http://www.freescale.com/files/microcontrollers/doc/data_sheet/MC68HC08JL3.pdf

Pág. 107 – 109 - Descripción Funcional

Pág. 115 – Interrupción de desborde del temporizador

Págs. 117 – 121. - Información de registros

2.2.9.3 Manual de Referencia del CPU

(a) http://www.freescale.com/files/microcontrollers/doc/ref_manual/CPU08RM.pdf

2.2.9.4 Página “web” sobre esta Nota Técnica

(a) <http://www.geocities.com/issaiass/>

2.2.10 Problemas Propuestos

2.2.10.1 Generar una Comparación de Salida (“Output Compare”) de veinte (20 ms) fija por el Canal 1, habilite las interrupciones del canal para sumar el siguiente ancho de la onda (pulso en bajo o alto).

2.2.10.2 Generar una Comparación de Salida (“Output Compare”) de mil (1000) ms fija por el Canal 0 y otra de quinientos (500) ms fija por el Canal 1, habilite tanto las interrupciones del Canal 0 y 1 para sumar el siguiente ancho de pulso.